

Calcul intégral (1)

Jean-Michel Sarlat

8 avril 2003

*Ce document a été généré par la **fabrique de Syracuse** à partir d'un fichier texte contenant des spécifications de contenu et des commandes d'interfaçage avec **Maxima**. Le traitement de ce fichier a donné lieu, dans le même temps que ce document, à la création d'une page html.*

URL : <http://melusine.eu.org/syracuse/maxima/>

1 Énoncé

Calculer les intégrales ou primitives suivantes :

1/ $\int_1^{27} \frac{1}{t\sqrt[3]{t}} dt$

2/ $\int_0^2 (1 - |x - 1|)^3 dx$

3/ $\int_0^4 \sqrt{2x + 1} dx$

4/ $\int_0^{1/2} x^3(1 - x^2)^{5/2} dx$

5/ $\int_0^1 \frac{1}{\sqrt{9x^2 + 3}} dx$

6/ $\int_0^{\pi/2} x^2 \sin x e^x dx$

7/ $\int \frac{\ln(\sin x)}{\cos^2 x} dx$

8/ $\int_0^{\pi/2} \frac{1}{3 \tan x + 2} dx$

9/ $\int_0^{\pi} \frac{1}{a^2 \cos^2 x + \sin^2 x} dx$

10/ $\int \frac{\cos 2x}{\sin x + \sin 3x} dx$ (CV : $t = \cos x$)

11/ $\int \frac{1 - \cos 2x}{\sin 3x} dx$ (CV : $t = \cos x$)

12/ $\int \frac{dx}{2 \operatorname{ch} x + \operatorname{sh} x + 1}$ (CV : $t = \operatorname{th} \frac{x}{2}$)

13/ $\int \frac{\sqrt{x+1} - \sqrt[3]{x+1}}{\sqrt{x+1} + \sqrt[3]{x+1}} dx$
(CV : $t = (x+1)^{1/6}$)

14/ $\int \frac{\sqrt{x^3+1}}{x} dx$ (CV : $t^2 = x^3 + 1$)

15/ $\int \frac{x+1}{\sqrt{x(1-2x)}} dx$ (CV : $x - \frac{1}{4} = \frac{1}{4} \cos t$)

16/ $\int \sin 2x \operatorname{sh} 3x dx$

17/ $\int \tan^5(x) dx$

18/ $\int \frac{\sin x}{2 + \tan^2 x} dx$

19/ $\int \frac{1}{\cos^4 x + \sin^4 x} dx$

2 Calculs

(C2) `load("integration.mc")$`

(C3) `integre(1/(t*t^(1/3)),t,1,27);`

(D3)
$$\int_1^{27} \frac{1}{t^{\frac{4}{3}}} dt = 2$$

(C4) `integre_chasles((1-abs(x-1))^3,x,0,1,2);`

(D4)
$$\int_0^2 (1 - |x - 1|)^3 dx = \frac{1}{2}$$

(C5) `integre(sqrt(2*x+1),x,0,4);`

(D5)
$$\int_0^4 \sqrt{2x + 1} dx = \frac{26}{3}$$

(C6) `integre(x^3*(1-x^2)^(5/2),x,0,1/2);`

(D6)
$$\int_0^{\frac{1}{2}} x^3 (1 - x^2)^{\frac{5}{2}} dx = \frac{2}{63} - \frac{45\sqrt{3}}{3584}$$

(C7) `integre(1/sqrt(9*x^2+3),x,0,1);`

(D7)
$$\int_0^1 \frac{1}{\sqrt{9x^2 + 3}} dx = \frac{\text{ASINH}(\sqrt{3})}{3}$$

(C8) `integre(x^2*sin(x)*exp(x),x,0,%pi/2);`

(D8)
$$\int_0^{\frac{\pi}{2}} x^2 e^x \sin x dx = \frac{(\pi^2 - 4) e^{\frac{\pi}{2}}}{8} + \frac{1}{2}$$

(C9) `primitive_simplifie(log(sin(x))/cos(x)^2,x, strig1, expand);`

(D9)
$$\int \frac{\log \sin x}{\cos^2 x} dx = -\tan x \log \sin x - x$$

(C10) `integre_limite(1/(3*tan(x)+2),x,0,%pi/2);`

$$(D10) \quad \int_0^{\frac{\pi}{2}} \frac{1}{3 \tan x + 2} dx = \frac{3 \log 3}{13} - \frac{3 \log 2}{13} + \frac{\pi}{13}$$

(C11) `block(assume(a>0),integre_chasles_limite(1/(a^2*cos(x)^2+sin(x)^2),x,0,%pi/2,%pi));`

$$(D11) \quad \int_0^{\pi} \frac{1}{\sin^2 x + a^2 \cos^2 x} dx = \frac{\pi}{a}$$

(C12) `primitive_simplifie(cos(2*x)/(sin(x)+sin(3*x)),x,strig2,radcan);`

$$(D12) \quad \int \frac{\cos(2x)}{\sin(3x) + \sin x} dx = -\frac{\cos x \log \tan\left(\frac{x}{2}\right) + 1}{4 \cos x}$$

(C13) `primitive_simplifie((1-cos(2*x))/sin(3*x),x,strig2,radcan);`

$$(D13) \quad \int \frac{1 - \cos(2x)}{\sin(3x)} dx = \frac{\log(2 \cos x + 1) - \log(2 \cos x - 1)}{2}$$

(C14) `primitive(1/(2*cosh(x)+sinh(x)+1),x);`

$$(D14) \quad \int \frac{1}{\sinh x + 2 \cosh x + 1} dx = \frac{2 \operatorname{Arctan}\left(\frac{6e^x+2}{2\sqrt{2}}\right)}{\sqrt{2}}$$

(C15) `f(x):=((x+1)^(1/2)-(x+1)^(1/3))/((x+1)^(1/2)+(x+1)^(1/3));`

$$(D15) \quad f(x) = \frac{\sqrt{x+1} - (x+1)^{\frac{1}{3}}}{\sqrt{x+1} + (x+1)^{\frac{1}{3}}}$$

(C16) `block(assume(t>0),primitive_cv(f(x),x,t-(1+x)^(1/6),t));`

$$(D16) \quad \int \frac{\sqrt{x+1} - (x+1)^{\frac{1}{3}}}{\sqrt{x+1} + (x+1)^{\frac{1}{3}}} dx = \int \frac{6t^6 - 6t^5}{t+1} dt$$

(C17) `rhs(%)=ev(rhs(%),nouns);`

$$(D17) \quad \int \frac{6t^6 - 6t^5}{t+1} dt = 12 \log(t+1) + \frac{5t^6 - 12t^5 + 15t^4 - 20t^3 + 30t^2 - 60t}{5}$$

(C18) `block(assume(t>0),primitive_cv(sqrt(x^3+1)/x,x,t^2-x^3-1,t));`

$$(D18) \quad \int \frac{\sqrt{x^3+1}}{x} dx = 2 \int \frac{t^2}{3t^2-3} dt$$

(C19) `rhs(%)=ev(rhs(%),nouns);`

$$(D19) \quad 2 \int \frac{t^2}{3t^2-3} dt = 2 \left(-\frac{\log(t+1)}{6} + \frac{\log(t-1)}{6} + \frac{t}{3} \right)$$

(C20) `primitive((x+1)/sqrt(x*(1-2*x)),x);`

$$(D20) \quad \int \frac{x+1}{\sqrt{(1-2x)x}} dx = \frac{5 \operatorname{Arcsin}(4x-1)}{4\sqrt{2}} - \frac{\sqrt{x-2x^2}}{2}$$

(C21) `primitive(sin(2*x)*sinh(3*x),x);`

$$(D21) \quad \int \sin(2x) \sinh(3x) dx = \frac{e^{-3x} ((3e^{6x}+3) \sin(2x) + (2-2e^{6x}) \cos(2x))}{26}$$

(C22) `primitive_cv(tan(x)^5,x,cos(x)-t,t);`

$$(D22) \quad \int \tan^5 x dx = - \int \frac{t^4 - 2t^2 + 1}{t^5} dt$$

(C23) `rhs(%)=ev(rhs(%),nouns);`

$$(D23) \quad - \int \frac{t^4 - 2t^2 + 1}{t^5} dt = -\log t - \frac{4t^2 - 1}{4t^4}$$

(C24) `primitive(sin(x)/(2+tan(x)^2),x);`

$$(D24) \quad \int \frac{\sin x}{\tan^2 x + 2} dx = \operatorname{Arctan} \cos x - \cos x$$

(C25) `primitive(1/(cos(x)^4+sin(x)^4),x);`

$$(D25) \quad \int \frac{1}{\sin^4 x + \cos^4 x} dx = \frac{\operatorname{Arctan} \left(\frac{2 \tan x + \sqrt{2}}{\sqrt{2}} \right)}{\sqrt{2}} + \frac{\operatorname{Arctan} \left(\frac{2 \tan x - \sqrt{2}}{\sqrt{2}} \right)}{\sqrt{2}}$$

3 Macros : integration.mc

```
/* integration.mc
   Quelques macros pour présenter des calculs d'intégrales.
   JM Sarlat - 2003 - http://melusine.eu.org/syracuse/maxima/
*/

/* Présentation d'une intégrale définie */
integre(f,v,a,b) := 'integrate(f,v,a,b) = integrate(f,v,a,b);

/* Intégration en appliquant la relation de Chasles */
integre_chasles(f,v,a,b,c) :=
    'integrate(f,v,a,c) = integrate(f,v,a,b) + integrate(f,v,b,c);

/* Intégration en utilisant un calcul de limite aux bornes */
integre_limite(f,v,a,b) := 'integrate(f,v,a,b) = ldefint(f,v,a,b);

/* Intégration à l'aide d'une relation de Chasles avec calcul de limite
   aux bornes */
integre_chasles_limite(f,v,a,b,c) :=
    'integrate(f,v,a,c) = ldefint(f,v,a,b) + ldefint(f,v,b,c);

/* Calcul de primitive */
primitive(f,v) := 'integrate(f,v) = integrate(f,v);

/* Macro de simplification d'une expression trigo (1) */
strig1(e,v) := ev(e,
    cos(v)+1=sin(v)/tan(v/2),
    tan(v/2)^2+1=2*tan(v/2)/sin(v),
    tan(v/2)^2-1=2*tan(v/2)/tan(v));

/* Macro de simplification d'une expression trigo (2) */
strig2(e,v) := block([i],i:trigreduce(e),ev(i,
    2*cos(v)+2=tan(v/2)^2*(2-2*cos(v)),
    cos(2*v)=2*cos(v)^2-1));

/* Mise en oeuvre de simplifications avec procédure finale */
primitive_simplifie(f,v,env,final) :=
    'integrate(f,v) = final(env(integrate(f,v),v));

/* Changement de variable */
primitive_cv(f,v,eq,t) :=
    'integrate(f,v) = changevar('integrate(f,v),eq,t,v);
```